



Ministério da Ciência e Tecnologia
Instituto Nacional de Pesquisas Espaciais



Análise Espacial de Dados Geográficos

Laboratório

Módulo: Análise Multi Critério

Referência	Banco de dados Piranga		Doc	LAB1_AMC.doc	
Autor	Eduardo C. G. Camargo	Versão	1.0	Data	DEZ / 2000
Revisão		Versão		Data	

Agradecimentos,

*a Edson Ricardo Soares Pereira Da Cunha, aluno de Mestrado em Sensoriamento Remoto,
pelo material cedido.*

ÍNDICE

1. INTRODUÇÃO
2. OBJETIVOS
3. FLUXOGRAMA DE TRABALHO
4. ATIVAR BANCO DE DADOS
5. MODELAGEM DO BANCO DE DADOS PIRANGA
6. CRIAR PROJETO
7. IMPORTAR DADOS
8. GERAÇÃO DE GRADE REGULAR PARA O PI TEORES_CROMO
9. GERAÇÃO DE GRADE REGULAR PARA O PI TEORES_COBALTO
10. GERAR MAPA PONDERADO DA GEOLOGIA
11. MAPEAR A GRADE DO PI TEORES_CROMO UTILIZANDO FUZZY LOGIC
12. MAPEAR A GRADE DO PI TEORES_COBALTO UTILIZANDO FUZZY LOGIC
13. CRUZAR OS PI'S CROMO_FUZZY, COBALTO_FUZZY E GEOLOGIA_PONDERADA UTILIZANDO A FUNÇÃO FUZZY GAMA.
14. CRIAR O PI CROMO_AHP UTILIZANDO AHP (PROCESSO ANALÍTICO HIERÁRQUICO).
15. REALIZAR O FATIAMENTO NO GEO-CAMPO GAMA_FUZZY
16. REALIZAR O FATIAMENTO NO GEO-CAMPO CROMO_AHP
17. ETAPA FINAL

APÊNDICES

MAPA GEOLÓGICO DE PIRANGA

EQUAÇÕES FUZZY PARA CROMO E COBALTO

FUZZY GAMA

REFERÊNCIAS BIBLIOGRÁFICAS

1- INTRODUÇÃO

Nesta introdução será abordado, de forma resumida, alguns aspectos importantes da prospecção mineral de Cromo usando técnicas de geoprocessamento.

Os dados que serão analisados neste laboratório foram obtidos a partir de trabalhos de campo realizados na região de Pinheiros Altos, município de Piranga, Minas Gerais. Região esta que está inserida no contexto geológico do Quadrilátero Ferrífero, região historicamente de grande importância mineira, caracterizada por um ambiente geológico favorável ao desenvolvimento de mineralizações auríferas e de outros metais, como cromo, cobre e zinco.

Analisou-se dados geológicos¹ e geoquímicos², sendo que os dados geológicos foram extraídos de mapas geológicos já existentes, resultado de trabalhos anteriores feitos na região e através de campanhas de campo onde foram feitos perfis geológicos³. Os dados geoquímicos foram obtidos a partir da coleta de 42 amostras dentro dos córregos e rios com o auxílio de bateia⁴, sendo estas posteriormente analisadas quimicamente pelo método de absorção atômica⁵.

Na análise do Mapa Geológico (composto de várias unidades caracterizadas por diversos tipos de rochas), foi dada maior importância às áreas com ocorrência de rochas ultramáficas⁶, pois nestas rochas se localizam as principais jazidas de cromo no mundo.

Para a região de estudo admite-se que a unidade litológica com maior potencial para ocorrências de cromo sejam as rochas do domínio de metaultrabásicas (mv1) do Complexo Metamórfico Santo Antônio do Pirapetinga, por serem estes litotipos ricos em cromo. Porém em perfis geológicos realizados na região também pôde-se observar a existência de corpos de rochas ultramáficas dentro da Unidade Média do Supergrupo Rio das Velhas e da Unidade Asap do Complexo Metamórfico Santo Antônio do Pirapetinga. Sendo esta observação muito importante na fase de determinação dos pesos de cada unidade no Mapa Geológico Ponderado.

¹ Tipos de rochas

² Valor de um determinado elemento analisado em uma amostra

³ Caminhamentos feitos no campo com o objetivo de determinar com mais detalhe as diversas variações de tipos de rochas

⁴ Equipamento utilizado por garimpeiros na procura de ouro

⁵ Método de análise química que objetiva a determinação dos teores de diversos elementos químicos presentes nas amostras

⁶ Rochas com alta concentração de minerais ricos em ferro e magnésio

Na análise dos dados geoquímicos considerou-se como importantes os teores de cromo (Cr) e cobalto (Co), o Cr por ser logicamente a principal evidência das melhores áreas para ocorrências dos depósitos e o Co por ser um indicativo indireto da presença de mineralizações de cromo, fato que em alguns pontos os altos valores de cromo estão associados aos altos valores de cobalto como pode ser visto no Gráfico 1.




Gráfico 1: Comparativo entre os teores de Cr & Co.

A associação do Cr e Co em rochas ultramáficas é um importante fator na seleção de áreas mais ou menos favoráveis. Partindo desta idéia os pontos onde os teores dos dois elementos se apresentam mais elevados são áreas mais favoráveis do que aquelas onde somente os teores de cromo estiverem elevados.


Importância da aplicação do geoprocessamento na pesquisa mineral:

Os projetos desenvolvidos em SIG aplicados a prospecção mineral tem o objetivo de combinar dados espaciais para descrever e analisar interações, de modo a fazer previsões através de modelos prospectivos empíricos e fornecer apoio nas decisões tomadas por especialistas, e consequentemente definir melhor as regiões mais enriquecidas num determinado bem mineral.


Dos mapas de potencialidade obtidos através de técnicas de integração e análise multicritério é possível observar um cenário com características geológicas importantes, que auxiliam na seleção de algumas áreas dentro da região de estudo, com parâmetros interessantes para a realização de uma pesquisa de maior detalhe, objetivando possíveis mineralizações de cromo. Neste tipo de trabalho pode-se então notar a importância da avaliação de favorabilidade utilizando um Sistema de Informação Geográfica (SIG) nas atividades de pesquisa mineral.

2- OBJETIVOS

Os objetivos deste trabalho são a seleção de áreas potenciais à prospecção de Cromo, a partir das técnicas AHP (Processo Analítico Hierárquico) e "Fuzzy Logic". Os dados foram obtidos através de campanhas de campo realizadas na região de Pinheiros Altos, município de Piranga, Minas Gerais, no período de Abril a Julho de 1996, numa área de 51,33Km².



3- FLUXOGRAMA DE TRABALHO



4. ATIVAR BANCO DE DADOS

Nome: Piranga

5. MODELAGEM DO BANCO DE DADOS PIRANGA

Atenção siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, sem acentuação, espaços, etc...

Nome da Categoria	Modelo
Amostras	MNT
Cromo_Fuzzy	MNT
Cobalto_Fuzzy	MNT
Gama_Fuzzy	MNT
Cromo_AHP	MNT
Geologia_Ponderada	MNT

Atenção siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, sem acentuação, espaços, etc...

Nome da Categoria	Modelo	Classes Temáticas	Visual/Cores
Drenagem	Temático	drenagens	Linha / BLUE_7
Recorte	Temático	cl_recorte	Linha / BLACK
Fatiamento	Temático	Alto Potencial	Área / RED_7
		Medio Potencial	Área / GREEN_3
		Baixo Potencial	Área / BLUE_1
		Background	Área / GRAY_1
Geologia	Temático	Asap - Sto Antonio Pirapetinga	Área / BLUE_7
		Arvs - Unidade Superior	Área / GREEN_1
		Granito - Granodiorito	Área / RED_7
		Arvm - Unidade Media	Área / YELLOW_4
		mv1 - Sto Antonio Pirapetinga	Área / BEIGE
		mb - Sto Antonio Pirapetinga	Área / BLUE_1

6. CRIAR PROJETO

Nome	Projeção	Modelo da Terra
Cromo	UTM	Hayford
Longitude de Origem: o 45 00 00		
Retângulo	X1= 675750.653	X2= 684240.837
envolvente	Y1= 7723706.593	Y2= 7732252.189

Ativar projeto Cromo.

7. IMPORTAR DADOS

Os arquivos a serem importados estão localizados no diretório “Dados”, que está sob o Banco de Dados “Piranga”.

Importar o arquivo Teores_Cobalto.spr

- Modelo: MNT
- Formato: ASCII
- Entidade: Amostra Unid: m
- Escala: 1/ 50000
- Resolução X: 30 Y: 30
- Projeto: Cromo
- Categoria: Amostras
- PI: Teores_Cobalto

Importar o arquivo Teores_Cromo.spr

- Modelo: MNT
- Formato: ASCII
- Entidade: Amostra Unid: m
- Escala: 1/ 50000
- Resolução X: 30 Y: 30
- Projeto: Cromo
- Categoria: Amostras
- PI: Teores_Cromo

Importar o arquivo Recorte.spr

- Modelo: Temático
- Formato: ASCII
- Entidade: linhas s/ ajustes Unid: m
- Escala: 1/ 50000
- Projeto: Cromo
- Categoria: Recorte
- PI: Limites

Selecionar PI Limites (Painel Controle)

- Menu->Temático *opção*: Edição Vetorial... (Ajustar, Poligonizar e Classificar)
- Menu->Temático *opção*: Vetor->Matiz... (Resol. X = 30 e Resol. Y =30)

Importar o arquivo Rede_Drenagem.spr

- Modelo: Temático
- Formato: ASCII
- Entidade: linhas s/ ajuste Unid: m
- Escala: 1/ 50000
- Projeto: Cromo
- Categoria: Drenagem
- PI: Rios

Importar o arquivo Mapa_Geologico_L2D.spr

- Modelo: Temático
- Formato: ASCII
- Entidade: linhas s/ ajuste Unid: m
- Escala: 1/ 50000
- Projeto: Cromo
- Categoria: Geologia
- PI: Mapa_Geologico

⚠ I M P O R T A N T E !

⚠ Após a importação do arquivo Mapa_Geologico_L2D.spr, Ajustar e Poligonalizar (*Criar a Topologia*).

Importar o arquivo Mapa_Geologico_LAB.spr

- Modelo: Temático
- Formato: ASCII
- Entidade: identif. Unid: m
- Escala: 1/ 50000
- Projeto: Cromo
- Categoria: Geologia
- PI: Mapa_Geologico

Selecionar PI Mapa_Geologico (Painel Controle)

- Menu->Temático *opção*: Vetor->Matiz... (Resol. X = 30 e Resol. Y =30)

Visualizar e verificar se todos os PI's foram importados corretamente.

8. GERAÇÃO DE GRADE REGULAR PARA O PI TEORES_CROMO

Menu -> MNT -> Geração de Grade Retangular...

Entidade: Amostra

PI de Saída: Teores_Cromo

Resolução: X(m): 30 Y(m): 30

Interpolador: Média Pond.

9. GERAÇÃO DE GRADE REGULAR PARA O PI: TEORES_COBALTO

Menu -> MNT -> Geração de Grade Retangular

Entidade: Amostra

PI de Saída: Teores_Cobalto

Resolução: X(m): 30 Y(m): 30

Interpolador: Média Pond.

10. GERAR MAPA PONDERADO DA GEOLOGIA

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Geologia_Ponderada

Pressione o botão Criar..., e edite o programa abaixo:

⚠ Atenção siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...

```
{
//Declaração
Tematico geo ("Geologia");
Numerico geoP ("Geologia_Ponderada");
Tabela geoT (Ponderacao);

//Instanciação
geo = Recupere (Nome="Mapa_Geologico");

geoP = Novo (Nome = "Geologia_Ponderada" , ResX = 30, ResY = 30,
            Escala = 50000, Min = 0, Max = 1);

geoT = Novo (CategoriaIni = "Geologia",
            "Granito - Granodiorito" : 0,
            "Arvs - Unidade Superior" : 0,
            "Arvm - Unidade Media" : 0.7,
            "mv1 - Sto Antonio Pirapetinga" : 1,
            "mb - Sto Antonio Pirapetinga" : 0.5,
            "Asap - Sto Antonio Pirapetinga" : 0.7);

//Operacao
geoP = Pondere (geo, geoT);
}
```

Salvar o programa editado

=> Antes de executar o programa, procure entende-lo !

Executar o programa Geologia_Ponderada

Observe o Painel de Controle !

11. MAPEAR A GRADE DO PI TEORES_CROMO UTILIZANDO FUZZY LOGIC.

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Cromo_Fuzzy

Pressione o botão Criar..., e edite o programa abaixo:

⚠ **Atenção** siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...

```
{
// Fuzzy cromo (ponto ideal com um teor de 1.855 % e ponto de cruzamento em 0.32)
//Declaração
Numerico cromo ("Amostras");
Numerico cromofuzzy ("Cromo_Fuzzy");

//Instanciação
cromo = Recuperar ( Nome= "Teores_Cromo" );
cromofuzzy = Novo (Nome = "Cromo_Fuzzy", ResX=30, ResY=30, Escala=50000, Min=0,
                    Max=1);

//Operação
cromofuzzy = (cromo < 0.20)? 0 : (cromo > 1.855)? 1 : 1/(1 + (0.424 * ((cromo -
                    1.855)^2)));
}
```

Salvar o programa editado

=> **Antes de executar o programa, procure entendê-lo !**

Executar o programa Cromo_Fuzzy. *Observe o Painel de Controle !*

12. MAPEAR A GRADE DO PI TEORES_COBALTO UTILIZANDO FUZZY LOGIC.

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Cobalto_Fuzzy

Pressione o botão Criar..., e edite o programa abaixo:

⚠ **Atenção** siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...

```
{
// Fuzzy cobalto (ponto ideal com um teor de 150.92 ppm e ponto de cruzamento em
// 80ppm)

//Declaração
Numerico cobal ("Amostras");
Numerico cobalfuzzy ("Cobalto_Fuzzy");

//Instanciação
cobal = Recuperar ( Nome= "Teores_Cobalto" );
cobalfuzzy = Novo( Nome = "Cobalto_Fuzzy" , ResX = 30, ResY = 30, Escala = 50000,
Min = 0, Max = 1 );

//Operação
cobalfuzzy=(cobal <60) ? 0 : (cobal>150.92) ? 1 : 1/( 1 +(0.000198*((cobal - 150.92 )^2
)));
}
```

Salvar o programa editado

=> Antes de executar o programa, procure entendê-lo !

Executar o programa Cobalto_Fuzzy. *Observe o Painel de Controle !*

13. CRUZAR OS PI'S CROMO_FUZZY, COBALTO_FUZZY E GEOLOGIA_PONDERADA UTILIZANDO A FUNÇÃO FUZZY GAMA.

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Gama_Fuzzy

Pressione o botão Criar..., e edite o programa abaixo:

⚠ **Atenção siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...**

```
{
//Declaração
Numerico cobal("Cobalto_Fuzzy"), cromo("Cromo_Fuzzy"), geol ("Geologia_Ponderada");
Numerico gama ("Gama_Fuzzy");

//Instanciação
cobal = Recupere (Nome= "Cobalto_Fuzzy");
cromo = Recupere (Nome= "Cromo_Fuzzy");
geol = Recupere (Nome= "Geologia_Ponderada");

gama=Novo (Nome="Gama_Fuzzy", ResX=30, ResY= 30, Escala=50000, Min=0, Max=1);

//Operação
g=0.70;
gama = (cobal*cromo*geol)^(1 - g) * ((1 - ((1 - cobal) * (1- cromo) * (1- geol)))^g);
}
```

Salvar o programa editado.

=> **Antes de executar o programa, procure entende-lo !**

Executar o programa Gama_Fuzzy. *Observe o Painel de Controle !*

14. CRIAR O PI CROMO_AHP UTILIZANDO AHP (PROCESSO ANALÍTICO HIERÁRQUICO).

Menu -> Análise -> Suporte à Decisão (AHP) ...

Crítério	Peso	Crítério
Cromo_Fuzzy	5	Melhor
Geologia_Ponder	2	Um Pouco Melhor
Geologia_Ponder	7	Muito Melhor
		Igual
		Igual
		Igual
		Igual
		Igual
		Igual
		Igual

Razão de Consistência: 0.012

Executando uma análise de suporte a decisão:

- **ative** o **Banco** e o **Projeto** que tem as definições no modelo de dados;
- **clique** em **Análise Espacial - Suporte à Decisão(AHP)...**;
- A janela associada a lista de **Categoria** apresenta somente as do modelo do banco que são **temáticas, numéricas** ou **imagem**. **Selecione** no **mínimo 2** e no **máximo 5** categorias. Caso selecione mais de cinco categorias, uma mensagem será apresentada informando que somente as cinco primeira serão consideradas;
- **clique** em **Exibir**. **Observe** que as categorias (comparação entre os diferentes critérios), duas a duas, serão apresentadas nos campos abaixo;
- **selecione** para cada par de categoria o **Peso** desejado. **Observe** que os valores correspondentes são apresentados à esquerda de cada botão;
- **clique** no botão **<=>** se desejar inverter a ordem entre cada par de categorias;
- **observe** ainda que o valor da **Razão de Consistência** é recalculado a cada alteração de peso. Caso o valor ultrapasse 0.1, será alertado antes de calcular os pesos para o programa a ser criado;
- **clique** em **Calcular Peso**. A janela de **Salvar Como** será apresentada para escolher o diretório onde será gravado o programa em **LEGAL**. **(Salvar no diretório Programas Legal sob o banco de dados Piranga, com o nome de Cromo_AHP).**

NOTA: Após selecionar quais os fatores que deseja combinar e estabelecer a importância relativa de cada um deles. O sistema fornecerá uma indicação da consistência de seu julgamento (indicada no item "razão de consistência").

Segundo os especialistas em AHP, é aconselhável que o índice de consistência seja sempre menor que 0,1. Assim, se seu índice de consistência foi maior que 0.1, considere a possibilidade de refazer seu julgamento.

Como resultado, esta função do SPRING gera um esqueleto de programa em LEGAL, que deverá ser completado pelo usuário com as informações específicas sobre os dados nos quais deseja aplicar o procedimento. Lembre-se que a aplicação da técnica AHP é sob a forma de uma média ponderada. Assim, os dados deverão ser convertidos para uma escala de [0..1] antes da aplicação do programa.

Edite o Programa, CromoAHP, gerado pela técnica AHP, conforme abaixo:

```
{
// Pesos a ser aplicados
// Cromo_Fuzzy = 0.333
// Cobalto_Fuzzy = 0.075
// Geologia_Ponderada = 0.592

// Razao de consistencia
// CR = 0.012

// Programa em LEGAL
// Este programa deve ser completado
// pelo usuario para incluir os dados
// apresentados entre os sinais de <> ← IMPORTANTE !

// Definicao dos dados de entrada

Numerico var1 ("Cromo_Fuzzy");
Numerico var2 ("Cobalto_Fuzzy");
Numerico var3 ("Geologia_Ponderada");

// Definicao do dado de saida

Numerico var4 ("<Categoria_de_saida>"); ← Cromo_AHP

// Recuperacao dos dados de entrada

var1 = Recupere (Nome="<Nome_do_PI>"); ← Cromo_Fuzzy
var2 = Recupere (Nome="<Nome_do_PI>"); ← Cobalto_Fuzzy
var3 = Recupere (Nome="<Nome_do_PI>"); ← Geologia_Ponderada

// Criacao do dado de saida

var4 = Novo (Nome="<Nome_PI_Saida>", ResX=<>, ResY=<>, Escala=<>,
             Min=0, Max=1);
             ↑           ↑           ↑           ↑
             Cromo_AHP   30         30         50000

// Geracao da media ponderada
var4 = 0.333*var1 + 0.075*var2+ 0.592*var3;
}
```

Salvar e Executar. Observe o Painel de Controle !

15. REALIZAR O FATIAMENTO NO GEO-CAMPO GAMA_FUZZY.

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Fatiamento_Gama_Fuzzy

Pressione o botão Criar..., e edite o programa abaixo:

⚠ **Atenção** siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...

```
{
//Declarações
Numerico num ("Gama_Fuzzy");
Tematico tem ("Fatiamento");
Tabela    tab(Fatiamento);

//Instanciações
num = Recupere (Nome = "Gama_Fuzzy");

tab = Novo (CategoriaFim = "Fatiamento",
            [0.0, 0.2] : "Background",
            [0.2, 0.5] : "Baixo Potencial",
            [0.5, 0.7] : "Medio Potencial",
            [0.7, 1.0] : "Alto Potencial" );

tem = Novo (Nome = "FAT_Gama_Fuzzy", ResX=30, ResY=30, Escala=50000);

//Operações
tem = Fatie (num, tab);
}
```

Salvar o programa editado.

=> **Antes de executar o programa, procure entende-lo !**

Executar o programa Fatiamento_Gama_Fuzzy. *Observe o Painel de Controle !*

16. REALIZAR O FATIAMENTO NO GEO-CAMPO CROMO_AHP.

Menu -> Análise -> Legal...

Diretório: vá para o diretório "Programas Legal" que está sob o Banco de Dados "Piranga".

Nome: Fatiamento_Cromo_AHP

Pressione o botão Criar..., e edite o programa abaixo:

⚠ **Atenção** siga exatamente os nomes abaixo, letras maiúsculas, minúsculas, espaços, sem acentuação, etc...

```
{
//Declarações
Numerico num ("Cromo_AHP");
Tematico tem ("Fatiamento");
Tabela    tab(Fatiamento);

//Instanciações
num = Recupere (Nome = "Cromo_AHP");

tab = Novo (CategoriaFim = "Fatiamento",
            [0.0, 0.2] : "Background",
            [0.2, 0.5] : "Baixo Potencial",
            [0.5, 0.7] : "Medio Potencial",
            [0.7, 1.0] : "Alto Potencial" );

tem = Novo (Nome = "FAT_Cromo_AHP", ResX=30, ResY=30, Escala=50000);

//Operações
tem = Fatie (num, tab);
}
```

Salvar o programa editado.

Antes de executar o programa, procure entende-lo !


Executar o programa Fatiamento_Cromo_AHP. *Observe o Painel de Controle !*

17. ETAPA FINAL

Apresente e Analise os Mapas de Potencialidade de Cromo gerados pelas técnicas AHP e Fuzzy Gama.

APÊNDICES

• MAPA GEOLÓGICO DE PIRANGA



TEOR DE CROMO NAS CLASSES GEOLÓGICAS


■	○ nada
■	○ muito bom
■	○ nada
■	○ ótimo
■	○ muito bom
■	○ bom

A associação dos pesos às unidades geológicas baseados na ocorrência de cromo resultaram nos seguintes valores:

- Granito-Granodiorito : 0
- Arvs - Unidade Superior : 0
- Arvm - Unidade Media : 0.7
- mv1 - Sto Antonio Pirapetinga : 1
- mb - Sto Antonio Pirapetinga : 0.5
- Asap - Sto Antonio Pirapetinga : 0.7


• EQUAÇÕES FUZZY PARA CROMO E COBALTO

Os valores de “Background” e Anomalia de 1ª ordem foram utilizadas na formulação da função quadrática (Equação 1), onde o “Background” foi considerado como valor do ponto de cruzamento e as anomalias de 1ª ordem de cada elemento foi considerada como valor 1, conforme ilustra a figura abaixo:



Equação 1:
$$r = \frac{1}{1 + a(z - b)^2}$$

O valor do parâmetro α para o Cobalto é $\sim 0,000198$ e para o Cromo $\sim 0,424$



- **FUZZY GAMA**

Este operador é definido por dois termos, um produto algébrico Fuzzy e uma soma algébrica Fuzzy, como segue:

$$m = (\text{soma algébrica Fuzzy})^g * (\text{produto algébrico Fuzzy})^{1-g}$$

No produto o operador faz a multiplicação dos membros dos diferentes planos de informação (Geo-Campos $[0,1]$), onde o valor de saída de um dado ponto é sempre menor ou igual ao valor do membro Fuzzy. Isso ocorre devido a multiplicação de valores iguais ou menores que 1. Já na soma algébrica o resultado é sempre maior ou igual ao valor de entrada do maior membro Fuzzy.

A importância maior ou menor do operador em cada termo (soma e produto) vai depender do valor atribuído para o expoente γ . Assim quando $\gamma=0$, o resultado dependerá apenas do termo produto algébrico Fuzzy, e quando $\gamma=1$, o resultado dependerá apenas do termo soma algébrica Fuzzy.

Exemplo em Legal:

:

$\gamma=0.70$;

$\mu = ((1 - ((1 - \text{cobal}) * (1 - \text{cromo})))^{\gamma}) * (\text{cobal} * \text{cromo})^{(1 - \gamma)}$;

:

REFERÊNCIAS BIBLIOGRÁFIAS

- Almeida, F. F. M. Províncias Estruturais Brasileiras. In: **Simpósio de Geologia do Nordeste**, 6, Campina Grande, PE, 1977. Ata p. 363-391.
- Bonham-Carter, G.F. – 1994 – **Geographic information systems for geoscientists, modelling with GIS**. Pergamon.
- Bonham-Carter, G.F. **Mapping mineral potential with a geographic information system**. Geological Survey of Canada, Ottawa, Ontario, Canada. 1990. 121-130P.
- Borrough, P. A. & McDonnell, R. A. – 1998 – **Principles of Geographic information systems**. Oxford University Press.
- Eastman, J. R.; Jin, W.; Kyem, P. A. K.; Toledano, J. Raster procedures for multi-criteria/multi-objective decisions. **Photogrammetric Engineering and Remote Sensing**, v.61, n.5, may, 1995. 539-547p.
- Machado Filho, L.; Ribeiro, M.; Gonzales, S.R.; Schenini, C.A.; Santos Neto, A.; Palmeira, R.C.; Pires, J.L.; Teixeira, W.; Castro, H.E.F. **Geologia das folhas Rio de Janeiro** (SF 23/24) escala 1:1.000.000, mapa e texto explicativo. Rio de Janeiro: RADAM Brasil- MME, 1983. 780p..
- Maranhão, R. J. L. **Introdução à pesquisa mineral**. Fortaleza: BNB. ETENE, 1982. 680p.
- Raposo, F. O. **Programa Levantamentos Geológicos Básicos do Brasil. folha Rio Espera**. SF. 23-X-B-IV. Belo Horizonte, MG: DNPM/CPRM, 1991. 200p..
- Saaty, T. L. **Multicriteria Decision Making – The Analytical Hierarchy Process**, Pittsburg, RWS Publications, 1992.